

Let SAS do the work for you: Tips and tricks for turning SAS output into client-ready tables

Becky Lien, Professional Data Analysts; Lily Dunk, Professional Data Analysts; Sara Richter, Professional Data Analysts

ABSTRACT

When putting together results tables for clients, how often have you thought, “There has to be a better way?” There is! Commonly used SAS® procedures like FREQ/TABULATE/MEANS/SURVEYFREQ provide a wide range of options to customize table output. Combine those options with ODS OUTPUT statements, and the customization possibilities are virtually endless. Utilizing these possibilities, we can prepare results in Excel without manually adjusting the output once in Excel.

In this brief session, we will show examples of tailored SAS output using ODS Excel destination and corresponding SAS syntax. Examples include using ODS OUTPUT statements within SAS procedures (FREQ/SURVEYFREQ/MEANS) and wrapping the procedure and data steps in a macro for easy reuse. We also utilize style sheets and conditional formatting to improve the look and readability of the finalized tables. Examples in this session were developed in Base SAS.

INTRODUCTION

As consultants in health services research and public health program evaluation, we have clients with a wide variety of statistical knowledge. We prefer to produce output in Microsoft Excel because many of our clients are Microsoft Office users and can quickly turn tabular results into charts or graphics using Excel. However, the standard SAS output is sometimes not easily understood by our clients. Therefore, we use ODS OUTPUT statements within SAS procedures to generate summarized datasets and PROC PRINT in all three examples in this paper to show the flexibility provided by this combination of features. All three examples also utilize the ODS output delivery system for Excel.

EXAMPLE 1: TAILORING PROC TTEST OUTPUT


In this example, output from PROC TTEST is sent to the ODS Excel destination after being reformatted and restructured to be more readable to our clients. The dataset SASHELP.CARS is used in this example. We want to report the results of a t-test comparing the miles per gallon (MPG) of cars in the USA to those in Europe.

DEFAULT OUTPUT

The following code sends the default output from PROC TTEST to an Excel file using the ODS EXCEL destination. By default, four separate tables are produced, and each table is output to a different Excel tab when using the ODS Excel destination (see Figure 1).

```
ODS EXCEL FILE='<filepath>\Example 1 - Before.xlsx';
PROC TTEST DATA=sashelp.cars;
  VAR MPG_City;
  CLASS origin;
  WHERE origin in ('USA', 'Europe');
RUN;
ODS EXCEL CLOSE;
```

Figure 1: PROC TTEST Default Output in ODS Excel Destination



Our goal is to replace these two tables with one table that includes: N, mean, standard deviation, and confidence limits.

We don't need the "Diff (1-2)" rows in either table.

From this table we will only report the Satterthwaite t-test results, limiting confusion for our clients by only presenting one option.

We won't report the information in this table.

TAILORING THE OUTPUT

To create these custom tables, we break up the code into two parts - the first part makes the combined mean summary table, and the second part prints the new summary table and the reduced t-test table.

Create a single mean summary table

```
PROC TTEST DATA=sashelp.cars;
  VAR MPG_City;
  CLASS origin;
  WHERE origin in ('USA', 'Europe');
  ❶ ODS OUTPUT Statistics=out_stats;
  ❷ ODS OUTPUT ConfLimits=out_CLs;
  ❸ ODS OUTPUT Ttests=out_ttests;
RUN;

PROC SORT DATA=out_stats;
  BY class;
RUN;
PROC SORT DATA=out_cls;
  BY class;
RUN;
❹ DATA MyDataTable;
  MERGE out_stats(KEEP=class N)
        out_cls (KEEP=class mean lowerclmean upperclmean stddev);
  BY class;
  ❺ WHERE CLASS in ('USA', 'Europe');
RUN;
```

TIP: To determine the available ODS OUTPUT table names for any SAS procedure, include “**ODS TRACE ON;**” before the procedure statement. The names of the available ODS OUTPUT tables are printed to the log file.

❶-❸: Sends default PROC TTEST output tables to separate datasets. The ODS table *Statistics* is saved as a SAS dataset called “out_stats” and contains all the information from the first table in Figure 1 (Ns, mean, standard deviation, minimum, and maximum). We will only use the Ns from this table. The ODS table *ConfLimits* is saved in a dataset called “out_CLs” and contains the information from the second table in Figure 1 (mean, standard deviation, and confidence limits). The ODS table *Ttests* is saved in a dataset called “out_ttests” and contains the information in the third table from Figure 1 (t-test statistic, degrees of freedom, and p-value for both a pooled test and a Satterthwaite test).

❹: Creates a dataset “MyDataTable” that combines “out_stats” and “out_CLs” into one table, merging on class. Uses KEEP statements to keep only the fields we want in our final table and the merging variable.

❺: Limits the rows in the dataset to keep the two classes (USA and Europe); this also excludes the rows “Diff (1-2)” seen in the default tables of Figure 1.

The “MyDataTable” dataset has the following rows and columns:

Class	N	Mean	LowerCLMean	UpperCLMean	StdDev
Europe	123	18.732	18.145	19.319	3.290
USA	147	19.075	18.426	19.724	3.983

Output the tailored summary table and t-test table

```
ODS EXCEL FILE='<filepath>\Example 1 Tailored.xlsx'
  ❶ OPTIONS (SHEET_INTERVAL='NONE' EMBEDDED_TITLES='ON'
  EMBED_TITLES_ONCE='ON' SHEET_NAME='MPG and Origin');

TITLE 'Mean MPG by origin';
```

```

❷ PROC PRINT DATA=MyDataTable NOOBS LABEL;
  ❸ VAR class N Mean StdDev LowerCLMean UpperCLMean;
  ❹ LABEL class='Program';
  LABEL stddev = 'Std Dev';
  LABEL LowerCLMean = 'Lower CI of Mean';
  LABEL UpperCLMean='Upper CI of Mean';
  FORMAT mean stddev LowerCLMean UpperCLMean 8.2;
RUN;

❺ PROC PRINT DATA=out_ttests NOOBS LABEL;
  WHERE method = 'Satterthwaite';
  VAR method tvalue df Probt;
  LABEL tvalue = 't statistic';
  LABEL probt = 'p-value';
RUN;
ODS EXCEL CLOSE;

```

❶: OPTIONS() in the ODS EXCEL statement are used to make four adjustments:

- SHEET_INTERVAL='NONE': keeps all procedure results on the same Excel tab. Without this, separate Excel tabs (worksheets) would be created for each of the PROC PRINT statements.
- EMBEDDED_TITLES='ON': prints the TITLE at the top of the Excel sheet. In this case, our TITLE='Mean MPG by origin' was set before to the ODS EXCEL statement.
- EMBED_TITLES_ONCE='ON': Titles are only displayed once instead of before each procedure.
- SHEET_NAME='MPG and Origin': names the Excel tab "MPG and Origin."

❷: Prints the newly created "MyDataTable" dataset, suppresses the observation numbers (NOOBS), and uses variable labels (instead of variable names) as column headings (LABEL).

❸: Lists the variables in the desired order to be printed.

❹: Creates custom labels for the column headings.

❺: Prints the t-test table, "out_ttests", created as part of the PROC TTEST ODS OUTPUT statement; includes a WHERE clause that limits the results to the Satterthwaite t-test.

Figure 2: Tailored PROC TTEST Output

Mean MPG by origin					
Program	N	Mean	Std Dev	Lower CI of Mean	Upper CI of Mean
Europe	123	18.73	3.29	18.14	19.32
USA	147	19.07	3.98	18.43	19.72

Method	t statistic	DF	p-value
Satterthwaite	-0.78	267.96	0.4389

This screenshot shows our final tailored output that omits redundant information, streamlines the information we did include, and hopefully makes the output easier for our clients to read.

EXAMPLE 2: TAILORING PROC FREQ OUTPUT WITH CHI-SQUARE

In this example, PROC FREQ output is tailored to be more readable to our clients. We use some tailoring presented in Example 1 and introduce some additional strategies, including using custom style sheets to have tables formatted in the desired company branding style. The full SAS code for this example can be found in the Appendix. We again use the SASHELP.CARS dataset. For this example, we want to report the results comparing the distribution of car origins (USA, Europe, Asia) by drive train (All, Front, Rear).

DEFAULT OUTPUT

The following code uses defaults from PROC FREQ and the ODS EXCEL output destination to produce an Excel file with two tables, one that displays the N and percent of each drive train category by car origin with row percentages. The other table includes Chi-square test information.

```
ODS EXCEL FILE='\\<filepath>\Example 2 - Before.xlsx';
  OPTIONS(SHEET_INTERVAL='NONE' EMBEDDED_TITLES='ON'
  EMBED_TITLES_ONCE='ON' SHEET_NAME='Car features by origin');
PROC FREQ DATA=SASHELP.CARS;
  TABLES drivetrain*origin/NOCOL NOPERCENT CHISQ;
RUN;
ODS EXCEL CLOSE;
```

Figure 3. Default PROC FREQ Output in ODS Excel Destination

The FREQ Procedure				
Table of DriveTrain by Origin				
DriveTrain	Origin			
Frequency Row Pct	Asia	Europe	USA	Total
All	34 36.96	36 39.13	22 23.91	92
Front	99 43.81	37 16.37	90 39.82	226
Rear	25 22.73	50 45.45	35 31.82	110
Total	158	123	147	428

Note: Statistics for Table of DriveTrain by Origin			
Statistic	DF	Value	Prob
Chi-Square	4	40.1784	<.0001
Likelihood Ratio Chi-Square	4	41.4879	<.0001
Mantel-Haenszel Chi-Square	1	3.5134	0.0609
Phi Coefficient		0.3064	
Contingency Coefficient		0.2929	
Cramer's V		0.2167	

Note: Sample Size = 428	
-------------------------	--

Our goal is to increase the readability of these default tables by updating:

- The default HTMLBlue style.
- The cells displaying N and row percentages together in the first table.
- No total percentages
- Extra table headings (e.g., “The FREQ Procedure”).
- The Note at the bottom of the page.
- Streamlining the multiple options for Chi-square tests as they can be confusing to those not as familiar with statistics.

TAILORING THE OUTPUT

This section walks through nine different revisions to the default code. The table below summarizes the revised approach. In each update sub-section, one or more update items are applied, and the resulting output is shown. Complete SAS code for the tailored version of Example 2 is in the Appendix and includes a macro that extends the code for different group/row variables.

Table 1. Example 2 Tailoring Summary

Issue with default output	Revised approach	Effect on output
Cells within table contain N and row percentage. Can't easily read.	Use PROC TABULATE for main table	Allows for more control in formatting the table summary, including: separating the N and row percentage cells, labeling, and control over which totals are displayed.
Percentages are not labeled with “%”	Apply a user-defined picture format with PROC FORMAT	Adds the % symbol to the ROWPCTN cells
HTMLBlue style can appear cluttered or busy	Specify a different style with ODS EXCEL: STYLE option	Uses one of many optional ODS EXCEL style templates available through SAS.
Key finding was not highlighted	PROC FORMAT: Conditional formatting	Highlights cells with ROWPCTN below 20% (or any threshold).
	ODS EXCLUDE	Suppress the PROC FREQ summary table with ODS EXCLUDE ALL and then displays the PROC PRINT table with ODS EXCLUDE NONE.
All Chi-square output was reported	PROC FREQ: ODS OUTPUT	Saves the Chi-square information from the PROC FREQ in a SAS dataset.
All Chi-square output was reported	PROC PRINT	Prints the information we want from the Chi-square table.
New Excel tab was created for each table	ODS EXCEL OPTIONS	Keeps both the summary table and Chi-square table on the same Excel tab. Controls the display of titles.
	Macro	Creates a custom Macro to wrap all of the tailoring syntax for ease of use with multiple group (row) variables. This macro is flexible for either character or numeric group variables and with or without formats applied.

Update 1: Prepare the table with PROC TABULATE and a picture format

```

PROC FORMAT;
  ❶ PICTURE pctfmt (ROUND) other='009.9%'; RUN;
ODS EXCEL FILE='\\<filepath>\Example 1 - Upd1_2.xlsx'
  OPTIONS(sheet_interval='NONE' EMBEDDED_TITLES='ON'
  EMBED_TITLES_ONCE='ON' SHEET_NAME='Car features by origin');
PROC TABULATE DATA=SASHELP.CARS;
  CLASS drivetrain origin;
  ❷ TABLE (drivetrain='' all='Total'),
    (origin='Origin' all='Total')*
    (N*format=comma8.0 ❸ ROWPCTN='%'*format=pctfmt.)/
  ❹ NOCELLMERGE ROW=float BOX='Drive train';
RUN;

```

```
ODS EXCEL CLOSE;
```

- ❶: Creates a picture format to be used with the ROWPCTN column in PROC TABULATE. This forces the % symbol to appear and forces one digit before and after the decimal (e.g. 0.1%, 1.0%, 1.1%, 10.1%.)
- ❷: The TABLE statement in PROC TABULATE creates separate cells for the N and the ROWPCTN.
- ❸: '=' re-labels the ROWPCTN column and format=pctfmt. applies the picture format to the ROWPCTN column.
- ❹: The PROC TABULATE TABLE statement, the options we've used include: NOCELLMERGE to separate the first row of the table from the header row; ROW=float to allow the row heading to be in line with the column headings instead of on a separate row; and BOX="Drive train" to re-label the row title.

Figure 4. Example 2 Tailoring Revisions Using PROC TABULATE and a Picture Format

Drive train	Origin							
	Asia		Europe		USA		Total	
	N	%	N	%	N	%	N	%
All	34	37.0%	36	39.1%	22	23.9%	92	100.0%
Front	99	43.8%	37	16.4%	90	39.8%	226	100.0%
Rear	25	22.7%	50	45.5%	35	31.8%	110	100.0%
Total	158	36.9%	123	28.7%	147	34.3%	428	100.0%

The resulting output is more user-friendly with Ns and row percentages in separate cells, and percentages now include the % symbol. A customized label “Drive train” is centered above the row headings for drive train.

Update 2: Change the ODS EXCEL Style and Apply Conditional Formatting

```
PROC FORMAT;
```

```
❶ VALUE cellColor low-19='yellow' other='white';
  PICTURE pctfmt (ROUND)other='009.9%';
```

```
RUN;
```

```
ODS EXCEL FILE='<filepath>\Example 1 - Updates 3 and 4.xlsx'
```

```
❷ STYLE=Word
```

```
  OPTIONS (SHEET_INTERVAL='NONE' EMBEDDED_TITLES='ON'
    EMBED_TITLES_ONCE='ON' SHEET_NAME='Car features by origin');
```

```
PROC TABULATE DATA=SASHELP.CARS;
```

```
  CLASS drivetrain origin;
```

```
  TABLE (drivetrain=' ' all='Total'), (origin='Origin' all='Total') *
    (N*format=comma8.0 ROWPCTN='% '*format=pctfmt.*
```

```
❸ [STYLE=[BACKGROUND=cellColor.]] /
```

```
  NOCELLMERGE ROW=float BOX='Drive train';
```

```
RUN;
```

```
ODS EXCEL CLOSE;
```

❶: Creates a conditional color format. When the value is less than 20, the color yellow is set. For all other values, white is set as the color. This will be used to format the background color of the ROWPCTN cells so that values less than 20% are highlighted in yellow.

❷: Applies the “Word” style to the ODS Excel output destination. SAS provides a number of options to create ODS EXCEL templates with different fonts, font weights, colors, and border options. We chose the “Word” style because we like the font and border selections and our clients often print in grayscale, so we don’t need color.

❸: Applies the conditional color formatting to the ROWPCTN cells.

Figure 5. Example 2 Style Sheet and Conditional Formatting

Drive train	Origin							
	Asia		Europe		USA		Total	
	N	%	N	%	N	%	N	%
All	34	37.0%	36	39.1%	22	23.9%	92	100.0%
Front	99	43.8%	37	16.4%	90	39.8%	226	100.0%
Rear	25	22.7%	50	45.5%	35	31.8%	110	100.0%
Total	158	36.9%	123	28.7%	147	34.3%	428	100.0%

After applying the style sheet, the Excel output is in our preferred style. In addition, the conditional formatting highlighted one cell where the row percent was less than 20% - making it easier to notice that front-wheel drives are less common in Europe.

Update 3: Tailor the Chi-square Table with ODS EXCLUDE, ODS OUTPUT, PROC PRINT, and ODS EXCEL OPTIONS

Updates 1 and 2 focused on preparing the data summary table, this next update prepares the Chi-square table and applies some final formatting.

```

❶ ODS EXCLUDE ALL;
PROC FREQ DATA=sashelp.cars;
  TABLES drivetrain*origin/NOPRINT CHISQ ;
❷ ODS OUTPUT ChiSq=OriginChiSq;
RUN;

❸ ODS EXCLUDE NONE;
ODS EXCEL FILE='\\<filepath>\Example 1 - Updates 5-7.xlsx'
  CSSSTYLE = "MyTables.css"
  OPTIONS(sheet_interval='none' EMBEDDED_TITLES='ON'
  EMBED_TITLES_ONCE='ON' SHEET_NAME='Origins');
TITLE 'Car origin analysis';
PROC TABULATE DATA=SASHELP.CARS;
  CLASS drivetrain origin;
  TABLE (drivetrain=' ' all='Total'), (origin='Origin' all='Total') *
    (N*format=comma8.0 ROWPCTN='% '*format=pctfmt.*
    [STYLE=[BACKGROUND=cellColor.]]) /
    NOCELLMERGE ROW=float BOX='Drive train';
RUN;
❹ PROC PRINT DATA=OriginChiSq LABEL NOOBS;
  VAR Value DF Prob;
  LABEL Value='Chi-sq statistic';
  LABEL Prob='P-value';
  WHERE Statistic = 'Chi-Square';

```

```

FORMAT value 8.2;
FORMAT prob 8.3;
RUN;
ODS EXCEL CLOSE;

```

- ❶: ODS EXCLUDE ALL suppresses the table results from PROC FREQ. Although this statement is not required in this code block as the PROC FREQ is outside the ODS EXCEL statement so it wouldn't be displayed in the Excel document. It is included here because it is used in the macro shown in the Appendix.
- ❷: ODS OUTPUT ChiSquare=OriginChiSq saves the Chi-square results table, *ChiSquare*, as a dataset named "OriginChiSq".
- ❸: ODS EXCLUDE NONE updates the setting so results are displayed.
- ❹: PROC PRINT statement only lists the Chi-square we care about; the options available in the ODS OUTPUT *ChiSquare* table include: Chi-Square, Likelihood Ratio Chi-Square, Mantel-Haenszel Chi-Square, Phi Coefficient, Contingency Coefficient, and Cramer's V. The PROC PRINT is used with a WHERE clause to limit the Chi-Square information to the one statistic (with DF and p-value) that we want.

Figure 6. Example 2 Final Updates

The screenshot shows an Excel spreadsheet with a pivot table titled 'Car origin analysis'. The pivot table has 'Drive train' as the row label and 'Origin' as the column label. The 'Origin' columns are further broken down into 'Asia', 'Europe', and 'USA', each with 'N' and '%' sub-columns. A 'Total' column is also present. The data is as follows:

Drive train	Origin							
	Asia		Europe		USA		Total	
	N	%	N	%	N	%	N	%
All	34	37.0%	36	39.1%	22	23.9%	92	100.0%
Front	99	43.8%	37	16.4%	90	39.8%	226	100.0%
Rear	25	22.7%	50	45.5%	35	31.8%	110	100.0%
Total	158	36.9%	123	28.7%	147	34.3%	428	100.0%

Below the pivot table is a summary table for the Chi-square statistic:

Chi-sq statistic	DF	P-value
40.18	4	0.000

These updates conclude tailoring the original PROC FREQ table summary with a Chi-square. Notice the title displayed at the top of the tab, the tab name, the highlighted cell, and the customized Chi-square table that only shows the Chi-square we want the client to use.

Update 4: Macro

A macro was created to extend the usefulness of this code, the complete SAS code for the macro is in the Appendix. The macro allows the group (row) variables – along with format and labels – to be passed in. The example code in the appendix for Example 2 includes a macro that works for either character or numeric variables and allows for a format to the group variable.

EXAMPLE 3: TAILORING PROC SURVEYFREQ OUTPUT FOR WEIGHTED DATA

In this example, ODS Excel output from PROC SURVEYFREQ is reformatted and restructured to be more readable to our clients. Many of the same update techniques used in Examples 1 and 2 are

employed here. For this example, we use the national 2020 BRFSS dataset, available at https://www.cdc.gov/brfss/annual_data/annual_2020.html. We want to report the smoking rates for Veterans and non-Veterans, along with Chi-square test results.

DEFAULT OUTPUT

The following code uses defaults from PROC SURVEYFREQ and ODS EXCEL output destination to produce an Excel file with our results.

```
PROC FORMAT;
  VALUE veteran_f 1='Yes' 2='No';
  VALUE rfsmoke_f 1='No' 2='Yes, current smoker';
RUN;
ODS EXCEL FILE='<filepath>\Example 3 Before.xlsx'
  OPTIONS(SHEET_INTERVAL='none');
PROC SURVEYFREQ DATA=brfss.llcp2020;
  TABLE veteran3*_RFSMOK3/chisq row cl;
  WEIGHT _llcpwt;
  WHERE veteran3 <7 and _RFSMOK3 <7;
  FORMAT veteran3 veteran_f. _RFSMOK3 rfsmoke_f.;
RUN;
ODS EXCEL CLOSE;
```

Figure 7: Example 3 Default Output

The screenshot shows an Excel spreadsheet with the following content:

Data Summary

Number of Observations	380061
Sum of Weights	243452177

Table of VETERAN3 by _RFSMOK3

VETERAN3	RFSMOK3	Frequency	Weighted Frequency	Std Err of Wgt Freq	Percent	Std Err of Percent	95% Confidence Limits for Percent	Row Percent	Std Err of Row Percent	95% Confidence Limits for Row Percent
Yes	No	39504	20956790	251577	8.6082	0.1005	8.4111 8.8052	84.1017	0.4085	83.3011 84.9024
	Yes, current smoker	6710	3961587	110930	1.6273	0.0453	1.5384 1.7161	15.8983	0.4085	15.0976 16.6989
	Total	46214	24918378	274153	10.2354	0.1089	10.0219 10.4489	100.0000		
No	No	288167	187846341	784303	77.1594	0.1547	76.8561 77.4627	85.9576	0.1354	85.6922 86.2229
	Yes, current smoker	45680	30687458	311576	12.6051	0.1222	12.3656 12.8446	14.0424	0.1354	13.7771 14.3078
	Total	333847	218533799	825757	89.7646	0.1089	89.5511 89.9781	100.0000		
Total	No	327671	208803132	810990	85.7676	0.1286	85.5156 86.0196			
	Yes, current smoker	52390	34649045	329766	14.2324	0.1286	13.9804 14.4844			
Total		380061	243452177	853451	100.0000					

Rao-Scott Chi-Square Test

Pearson Chi-Square	98.5229
Design Correction	4.8886
Rao-Scott Chi-Square	20.1536
DF	1
Pr > ChiSq	<.0001
F Value	20.1536
Num DF	1
Den DF	380060
Pr > F	<.0001

Sample Size = 380061

The code above produces an Excel spreadsheet with three tables. The format of the main data table with smoking rates has the most readability issues for our clients. We also only need the Rao-Scott Chi-square from the last table.

TAILORING UPDATES

This section creates tailored output for this analysis in three steps: saving the ODS OUTPUT we need, restructuring the smoking rate table, and exporting to Excel.

Save ODS OUTPUT

```
PROC SURVEYFREQ DATA=brfss.llcp2020;
  TABLES veteran3*_RFSMOK3/CHISQ ROW CL;
  WEIGHT _llcpwt;
  ❶ ODS OUTPUT CrossTabs=xTabsSmoke;
  ❷ ODS OUTPUT ChiSq=chisqSmoke;
  FORMAT veteran3 veteran_f. _RFSMOK3 rfsmoke_f.;
  WHERE veteran3 <7 and _RFSMOK3 <7;
RUN;
```

❶: Saves the ODS OUTPUT CrossTabs dataset as “xTabsSmoke”.

❷: Saves the ODS OUTPUT ChiSq dataset as “ChiSqSmoke”.

The fields we'll use from the ODS OUTPUT *CrossTabs* table (xTabsSmoke) are below. We will only use the highlighted data from observations 2, 3, 5, and 6.

Obs	F_VETERAN3	_RFSMO K3	Frequency	WgtFreq	RowPercent	RowLowerCL	RowUpperCL
1	Yes	1	39504	20956790.5	84.102	83.3011	84.9024
2	Yes	2	6710	3961587.43	15.898	15.0976	16.6989
3	Yes	.	46214	24918377.9	100	–	–
4	No	1	288167	187846341	85.958	85.6922	86.2229
5	No	2	45680	30687457.8	14.042	13.7771	14.3078
6	No	.	333847	218533799	100	–	–
7	Total	1	327671	208803132	–	–	–
8	Total	2	52390	34649045.2	–	–	–
9	Total	.	380061	243452177	–	–	–

Restructure smoking rate table

```
❶ DATA smoketable_yes smoketable_total;
  SET xtabssmoke;
  WHERE F_veteran3 in ('Yes', 'No');
  IF _RFSmok3 EQ 2 THEN OUTPUT smoketable_yes;
  ELSE IF _RFSmok3 EQ . THEN OUTPUT smoketable_total;
RUN;
DATA smoketable_total2;
  SET smoketable_total;
  ❷ RENAME Frequency=TotN_uwt WgtFreq=TotN_wt;
RUN;
PROC SORT DATA=smoketable_yes;
  BY F_Veteran3;
RUN;
PROC SORT DATA=smoketable_total2;
  BY F_Veteran3;
```

```

RUN;
❶ DATA smoketable_final;
  MERGE smoketable_yes
        smoketable_total2 (KEEP=F_veteran3 totN_uwt TotN_wt);
  BY F_Veteran3;
RUN;

```

- ❶: Creates two datasets from the “xTabsSmoke” dataset, one for the smoker rows (_RFSMOK3=2) and one for the total rows (_RFSMOK3=.). The WHERE clause excludes the F_Veterans='Total' rows.
- ❷: In our new “smoketable_total” dataset, we rename the Frequency and WgtFreq variables.
- ❸: Merges the two rows in the “smoketable_yes” dataset with the two rows in “smoketable_total2”.

Send smoking rate table and Chi-square table to Excel

```

TITLE 'Current smoking rate';
ODS EXCEL FILE='<filepath>\Example 3 Tailored.xlsx'
  STYLE=Word OPTIONS(SHEET_INTERVAL='none' EMBEDDED_TITLES='ON'
  EMBED_TITLES_ONCE='ON' SHEET_NAME='Current smokers');
PROC PRINT DATA=Smoketable_final LABEL NOOBS;
  VAR F_Veteran3 TotN_uwt TotN_wt Frequency WgtFreq RowPercent
  RowLowerCL RowUpperCL;
  LABEL F_Veteran3 = 'Veteran Status';
  LABEL TOTN_uwt = 'Total N (unwt)';
  LABEL TotN_wt='Total N (wt)';
  LABEL N_uwt='Current smoker N (unwt)';
  LABEL N_wt='Current smoker N (wt)';
  LABEL Pct_wt = 'Current smoking rate (wt)';
  LABEL lowerCI_pct_wt = 'Lower CI of smoking rate';
  LABEL upperCI_pct_wt = 'Upper CI of smoking rate';
  FORMAT RowPercent RowLowerCL RowUpperCL pctfmt.;
  FORMAT TotN_uwt TotN_wt Frequency WgtFreq comma8.0;
RUN;
PROC PRINT DATA=chisqSmoke LABEL NOOBS;
  WHERE Namel in ('_RSCHI_', 'DF_RSCHI', 'P_RSCHI');
  VAR Label1 cValue1;
  LABEL label1='Chi-sq statistics' cvalue1='value';
RUN;
ODS EXCEL CLOSE;

```

Figure 8: Example 2 with Tailoring

The screenshot shows an Excel spreadsheet with the following data:

Current smoking rate								
Veteran Status	Total N (unwt)	Total N (wt)	Current smoker N (unwt)	Current smoker N (wt)	Current smoking rate (wt)	Lower CI of smoking rate	Upper CI of smoking rate	
Yes	46,214	24,918,378	6,710	3,961,587	15.9%	15.1%	16.7%	
No	333,847	218,533,799	45,680	30,687,458	14.0%	13.8%	14.3%	

Chi-sq statistics	value
Rao-Scott Chi-Square	20.1536
DF	1
Pr > ChiSq	<.0001

The final output has only the information we need and is formatted to our specifications.

CONCLUSION

Using ODS OUTPUT to save results tables from SAS procedures provides excellent flexibility when sending SAS output to Microsoft Excel using the ODS EXCEL output destination. When combined with a few DATA steps, conditional formatting, and PROC PRINT, your SAS output won't need annotating or manual reformatting before sending to clients.

APPENDIX: SAMPLE CODE

FULL CODE FOR EXAMPLE 2:

```

/* CREATE FORMATS */
PROC FORMAT;
  /* conditional formatting to highlight cells with <20% */
  VALUE cellColor
    LOW-19='CXE6A22E'
    20-HIGH='white';

  /* picture format adds % symbol to the PROC TABULATE % cells */
  PICTURE pctfmt (ROUND)other='009.9%';

  /*format cylinders: 3-5, 6,8,10+*/
  VALUE cyl_f
    3-5='3 to 5'
    6='6'
    8='8'
    10-HIGH='10 to 12';
RUN;

/* WRAP ANALYSIS AND OUTPUT IN A MACRO */
/* PARAMETERS:

```

```

    ROWVAR = the variable represented in rows (e.g., drivetrain)
    ROWFMT = format for the rowvar, if no format needed can omit
    ROWHEADING = label for the row variable that will display
                 centered above the row labels */
%MACRO RowTblChiSq (rowvar, rowfmt, rowHeading);

/* Use ODS EXCLUDE NONE to make sure PROC TABULATE is displayed */
ODS EXCLUDE NONE;
PROC TABULATE DATA=SASHELP.cars;
  CLASS &rowvar origin;
  TABLE (&rowvar=' ' ALL='Total'), (origin='Origin'
    ALL='Total') * (N*format=comma8.0
    ROWPCTN='% '*format=pctfmt.*[STYLE=[BACKGROUND=cellColor.]])
    /ROW=float BOX="&rowHeading." NOCELLMERGE PRINTMISS
    MISSTEXT='0';
  FORMAT &rowvar &rowfmt;
RUN;

/* Use ODS EXCLUDE ALL to suppress the PROC FREQ results (but the
ODS OUTPUT dataset is still created) */
ODS EXCLUDE ALL;
PROC FREQ DATA=SASHELP.cars;
  TABLES &rowvar*origin/NOPRINT CHISQ;
  ODS OUTPUT ChiSq=OriginChiSq;
  FORMAT &rowvar &rowfmt;
RUN;

/* Use ODS EXCLUDE NONE to make sure PROC PRINT is displayed */
ODS EXCLUDE NONE;
PROC PRINT DATA=OriginChiSq LABEL NOOBS;
  VAR Value DF Prob;
  LABEL Value='Chi-sq statistic';
  LABEL Prob='P-value';
  WHERE Statistic = 'Chi-Square';
  FORMAT value 8.2;
  FORMAT prob 8.3;
RUN;

%MEND RowTblChiSq;

/** Run macro and dump results to an Excel spreadsheet */
ODS EXCEL FILE='<filepath>\Example 2 Tailored.xlsx'
  STYLE=Word
  OPTIONS(SHEET_INTERVAL='none' EMBEDDED_TITLES='ON'
  EMBED_TITLES_ONCE='ON' SHEET_NAME='Car features and origins');
TITLE 'Car origin analysis';
/*Pass in the character variable drivetrain with no format */
%RowTblChiSq(drivetrain,,Drive train);
/*Pass in the numeric variable cylinders along with format cyl_f.*/
%RowTblChiSq(cylinders,cyl_f.,Number of cylinders);
ODS EXCEL CLOSE;

```

REFERENCES

Benjamin, W.E. Jr. 2019. "SAS ODS Destination for Microsoft Excel: Use the STYLE Option to Spruce Up an Excel Output Workbook." Proceedings of the SAS Global 2019 Conference, Dallas TX: SAS Institute Inc. Available at <https://www.sas.com/content/dam/SAS/support/en/sas-global-forum-proceedings/2019/3775-2019.pdf>

SAS Institute Inc. 2017. "ODS OUTPUT: Store any statistic created by any SAS procedure" Accessed August 10, 2022. <https://blogs.sas.com/content/iml/2017/01/09/ods-output-any-statistic.html>

SAS Institute Inc. 2021. "ODS Table Names Produced by Base SAS Procedures." Accessed August 5, 2022. https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/odsproc/p037wkiv6e4hqIn1snmfk9b7c9it.htm

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Becky Lien
Professional Data Analysts
612-623-9110
blen@pdastats.com
<https://pdastats.com>